

Microsoft Visual Basic 6.0

Mit dem Multimedia-MCI-Steuererelement Audio- und Videodaten einbinden

Das ActiveX-Steuererelement *Multimedia-MCI (MMControl)* ist Bestandteil von Microsoft Multimedia Control 6.0 (mci32.ocx). Um dieses Steuererelement in Ihrem Programm nutzen zu können, müssen Sie es über den Befehl *Komponenten* aus dem Menü *Projekt* in die Werkzeugsammlung aufnehmen. Dann klicken Sie auf das Steuererelement und erstellen die Befehlsleistenoberfläche für das Formular. Das Multimedia-MCI-Steuererelement besteht aus mehreren Befehlsschaltflächen, die automatisch funktionieren, wenn ein gültiges Multimedia-Gerät geöffnet und das Steuererelement aktiviert ist. Die Schaltflächen tragen die Bezeichnungen *Zurück, Nächstes, Start, Pause, Rücklauf, Schritt, Stop, Aufnahme* und *Auswerfen*. Sie können diese Schaltflächen mit besonderen Funktionen ausstatten, indem Sie Ereignisprozeduren für bestimmte Schaltflächenereignisse erstellen. Normalerweise ist jedoch keine zusätzliche manuelle Konfiguration der Schaltflächen notwendig. Bei den meisten Anwendungen sind die Standardeinstellung der Schaltflächen zum Abspielen von Musik- und Videosequenzen völlig ausreichend.

Das Multimedia-MCI-Steuererelement kann in verschiedener Weise eingesetzt werden. Das Steuererelement kann während der Programmausführung sichtbar sein (Voreinstellung), und Sie können es verwenden, um dem Anwender eine komfortable Möglichkeit zur Handhabung eines angeschlossenen Multimedia-Geräts zu bieten, wie z.B. eines Videorecorders oder eines Audio-CD-Spielers. Sie können das Steuererelement während der Programmausführung auch ausblenden, indem Sie dessen Eigenschaft *Visible* den Wert *False* zuweisen. Diese Art der Verwendung ist besonders nützlich, wenn Sie das Steuererelement zur Wiedergabe von Audio- oder Spezialeffekten in Programmen einsetzen möchten. Beide Verwendungsmöglichkeiten werden in dieser Lektion beschrieben.

Die Eigenschaft *DeviceType*

Bevor Sie die Schaltflächen des Multimedia-MCI-Steuererelements benutzen können, müssen Sie ein gültiges Multimedia-Gerät über die Eigenschaft *DeviceType* des Steuererelements öffnen. Normalerweise wird hierzu entsprechender Programmcode in die Ereignisprozedur *Form_Load* eingefügt, damit das Steuererelement beim Programmstart automatisch konfiguriert wird. Die Eigenschaft *DeviceType* kann auch zur Laufzeit geändert werden, wenn mit einem Steuererelement verschiedene Multimedia-Geräte verwaltet werden sollen.

Die Syntax der Eigenschaft *DeviceType* lautet wie folgt:

```
MMControl1.DeviceType = DevName
```

Hierbei ist *DevName* eine Zeichenfolgenkonstante, die einen der gültigen Gerätetypen bezeichnet. Um beispielsweise ein Gerät anzugeben, das .wav-Dateien (WaveAudio) abspielen kann, müssten Sie folgende Zeichenfolge eingeben:

```
MMControl1.DeviceType = "WaveAudio"
```

Die folgende Tabelle 17.1 gibt einen Überblick über die aktuell vom Multimedia-MCI-Steuererelement unterstützten Multimedia-Geräten und die

entsprechenden *DevName*-Argumente für die Eigenschaft *DeviceType*:

Die Eigenschaft *Command*

Nachdem Sie mit der Eigenschaft *DeviceType* das Gerät angegeben haben, das verwendet werden soll, können Sie beginnen, über die Eigenschaft *Command* MCI-Befehle an das Gerät zu senden. Die verfügbaren Befehle entsprechen den englischen Namen der Schaltflächen des Multimedia-MCI-Steurelements: *Prev*, *Next*, *Play*, *Pause*, *Back*, *Step*, *Stop*, *Record* und *Eject*. Zudem kann das Steuerelement bestimmte allgemeine MCI-Befehle verarbeiten, wie *Open*, *Close*, *Sound*, *Seek* und *Save*. Die Verwendung von MCI-Befehlen ist besonders praktisch, da sie keine besonders detaillierten Kenntnisse der betreffenden Multimedia-Geräte

Multimedia-Gerät	DevName-Konst.	Beschreibung
Video (.avi-Dateien)	AVIVideo	Microsoft AVI-Format Video
Audio-CDsCDAudio	Musik-CDs	Musik ü. CD-ROM-Laufwerk
DAT-Band	DAT	Angeschlossenes DAT-Bandlaufwerk
Digitales Video	DigitalVideo	Digitale Videodaten
Video	MMMovie	Multimedia Movie-Format
-- (Das Video wird in einem Fenster angezeigt.)		
Video	Overlay	Frame-Overlay-Gerät
(Das Video wird in einem Fenster angezeigt.)		
Scanner	Scanner	Angeschlossener Scanner
MIDI-Sequencer	Sequencer	MIDI-Sequencer-Daten
Videoband	VCR	Angeschlossener Videorecorder
(Das Video wird in einem Fenster angezeigt.)		
Videodisc	Videodisc	Angeschlossenes Videodisc-Abspielgerät
Wave (.wav-Dateien)	WaveAudio	Microsoft Windows Audiodatei
Benutzerdefiniert	Other	Benutzerdefinierter Multimediatyp

erfordern. Beispielsweise ist es nicht notwendig, dass Sie genau wissen, wie die Daten auf einer CD im CD-ROM-Laufwerk gespeichert werden, um einen Befehl zu senden, mit dem die CD abgespielt wird.

In der folgenden Programmanweisung wird die Eigenschaft *Command* des *Multimedia-MCI-Steurelements* (*MMControl*) verwendet, um das Medium im geöffneten Multimedia-Gerät abzuspielen:

`MMControl1.Command = "Play"`

Das Programm *PlayTune*

Verschiedene kommerzielle Softwareprogramme (z.B. Microsoft Windows) spielen ein musikalisches „Leitmotiv“ ab, wenn sie geöffnet werden.

Mit dem Multimedia-MCI-Steurelement können Sie eine .wav-Datei, die einen Audioeffekt oder ein Lied enthält, beim Programmstart abspielen lassen. Das Dateiformat .wav (WaveAudio) ist ein verbreiteter Multimedia-Standard, der von der Firma Microsoft verwendet wird, um Audiodaten zu speichern. Sie können mit verschiedenen im Handel erhältlichen Audioprogrammen (wie z.B. Sound Forge) Dateien im .wav-Format erzeugen und bearbeiten. Sie finden .wav-Dateien aber auch im Lieferumfang des Betriebssystems Microsoft Windows und in Anwendungspaketen wie Microsoft Office.

Das Programm *PlayTune* ausführen

In der folgenden Übung werden Sie das Programm *PlayTune* ausführen und die Datei *applause.wav* mit dem Multimedia-MCI-Steurelement abspielen.

1 Starten Sie Visual Basic, und öffnen Sie das Projekt *PlayTune.vbp* aus

dem Ordner `|Vb6SfS\Lekt17`.

2 Klicken Sie in der Symbolleiste auf die Schaltfläche *Starten*, um das Programm auszuführen.

Visual Basic zeigt den Startbildschirm (das Startformular) des Programms an (siehe Abbildung 17.2) und spielt die Datei *applause.wav* mit dem Multimedia-MCI-Steuerelement ab. Das Startformular wird Ihnen möglicherweise bekannt vorkommen. Es wurde bereits in Lektion 9 verwendet, um zu zeigen, wie Sie ein zusätzliches Formular laden können, um einführende Informationen zu einem Programm anzuzeigen. In diesem Beispiel dient das Startformular lediglich als Blickfang, während die *.wav*-Datei wiedergegeben wird.

3 Klicken Sie im Formular auf die Schaltfläche *Weiter*, nachdem der Applaus endet.

Da dieses Startformular nur zu Demonstrationszwecken dient, wird das Programm nun beendet.

Im folgenden Abschnitt wollen wir uns den Programmcode ansehen, mit dem dieser Audioeffekt erzeugt wurde.

Der Programmcode zum Abspielen von *.wav*-Dateien

Das Programm *PlayTune* verwendet ein Multimedia-MCI-Steuerelement zum Abspielen der Datei *applause.wav*, wenn das Formular geladen wird. Das Multimedia-MCI-Steuerelement wird allerdings nicht angezeigt. Stattdessen wird in diesem Beispiel mit Hilfe des Multimedia-MCI-Steuerelements ein Spezialeffekt im Programm erzeugt, ohne die Befehlsschaltflächen des Steuerelements zu verwenden. Um das Multimedia-MCI-Steuerelement auszublenden, wurde seine Eigenschaft *Visible* im Eigenschaftenfenster auf *False* eingestellt.

1 Öffnen Sie die Ereignisprozedur *Form_Load* des Programms *PlayTune* im Codefenster.

Es wird folgender Programmcode angezeigt:

```
Private Sub Form_Load()  
MMControl1.Notify = False  
MMControl1.Wait = True  
MMControl1.Shareable = False  
MMControl1.DeviceType = "WaveAudio"  
MMControl1.FileName = "c:\Vb6SfS\Lekt17\applause.wav"  
MMControl1.Command = "Open"  
MMControl1.Command = "Play"  
End Sub
```

Damit beim Programmstart ein musikalisches „Leitmotiv“ abgespielt wird, verwenden Sie die Ereignisprozedur *Form_Load*, um der Eigenschaft *DeviceType* des Multimedia-MCI-Steuerelements den Wert *WaveAudio* (für *.wav*-Dateien) zuzuweisen und über die Eigenschaft *Command* die beiden MCI-Befehle *Open* und *Play* auszuführen. Neben diesen wichtigen Eigenschaften wurden einige andere Eigenschaften eingestellt, um das Multimedia-MCI-Steuerelement auf typische Laufzeitergebnisse vorzubereiten. Diese Einstellungen sind relativ allgemein und lassen sich in vielen Anwendungen einsetzen. Beachten Sie die folgenden Einstellungen:

■ Der Eigenschaft *Notify* wurde der Wert *False* zugewiesen, damit das Multimedia-MCI-Steuerelement keine Benachrichtigung ausgibt, nachdem die Befehle *Open* und *Play* ausgeführt worden sind. (Der Wert *True* erzeugt ein sogenanntes Callback-Ereignis (Rückruf), das

in diesem relativ einfachen Programm nicht relevant ist.)

n Der Eigenschaft *Wait* wurde der Wert *True* zugewiesen, da das Steuerelement warten soll, bis der Befehl *Open* aufgeführt wurde, bevor der Befehl *Play* an das Gerät gesendet wird.

n Die Eigenschaft *Shareable* des Multimedia-MCI-Steuerelements wurde auf *False* eingestellt, damit andere Anwendungen im System nicht auf das geöffnete MCI-Gerät zugreifen können. (Damit soll verhindert werden, dass zwei Anwendungen verschiedene Teile derselben Audiodatei gleichzeitig abspielen.)

n Der Eigenschaft *FileName* wurde der Pfad der Datei *applause.wav* als Wert zugewiesen. In den meisten Fällen müssen Sie die Eigenschaft *FileName* verwenden, um die wiederzugebende Multimedia-Datei anzugeben, bevor das Gerät mit dem Befehl *Open* geöffnet werden kann. Eine wichtige Ausnahme bildet die Wiedergabe von Audio-CDs, die sich in einem angeschlossenen CD-ROM-Laufwerk befinden. In diesem Fall muss kein Dateiname angegeben werden, da der Inhalt von Audio-CDs nicht über den Dateinamen verwaltet wird. Zur Wiedergabe von Audio-CDs genügt es, der Eigenschaft *Command* den Befehl *Open* zuzuweisen, um das erste Lied der CD abzuspielen.

2 Öffnen Sie die Ereignisprozedur *Form_Unload* im Codefenster.

Es wird folgender Programmcode angezeigt:

```
Private Sub Form_Unload(Cancel As Integer)
```

```
MMControl1.Command = "Close"
```

```
End Sub
```

Wenn das Multimedia-MCI-Steuerelement nicht mehr benötigt wird, sollten Sie der Eigenschaft *Command* den Befehl *Close* zuweisen, damit die vom Multimedia-Gerät belegten Systemressourcen wieder freigegeben werden. Diese Anweisung wird am besten in der Ereignisprozedur *Form_Unload* untergebracht, da diese Prozedur bei jeder *normalen* Beendigung des Programms (das heißt ohne abzustürzen) ausgeführt wird.

Durch die Aufnahme des Befehls *Close* in die Prozedur *Form_Unload* wird insbesondere den Anwendern Rechnung getragen, die die Anwendung durch Klicken auf die Schaltfläche *Schließen* in der Titelleiste beenden, da damit ebenfalls die Ereignisprozedur *Form_Unload* ausgelöst wird.

Video aus .avi-Dateien wiedergeben

Eine weiterer interessanter Spezialeffekt, den Sie in Programme einbauen können, ist die Wiedergabe eines Videoclips aus einer .avi-Datei.

.avi-Dateien haben ein Standarddateiformat zum Speichern von Videoaufzeichnungen mit Ton. Um das Multimedia-MCI-Steuerelement zum

Abspielen von .avi-Dateien zu konfigurieren, müssen die Eigenschaften *DeviceType*, *FileName* und *Command* wie folgt eingestellt werden:

```
MMControl1.DeviceType = "AVIVideo"
```

```
MMControl1.FileName = "c:\vb6sfs\lekt17\michael.avi"
```

```
MMControl1.Command = "Open"
```

```
MMControl1.Command = "Play"
```

Wenn Sie diese Programmanweisungen in ein Programm mit einem aktivierten Multimedia-MCI-Steuerelement einfügen, lädt das Steuerelement die angegebene Videodatei (in diesem Beispiel michael.avi) und spielt sie ab.

Wenn Sie ein Video im Format .avi wiedergeben, wird es in einem eigenen Fenster angezeigt und automatisch abgespielt. Es muss kein spezielles Formular für das Video erstellt werden, und die Wiedergabe muss

auch nicht über das Programm gesteuert werden.

Das Programm RunVideo

Das Programm *RunVideo* zeigt, wie das Multimedia-MCI-Steuerelement eingesetzt wird, um ein Video aus einer .avi-Datei in einer Visual Basic-Anwendung abzuspielen. Um das Programm etwas nützlicher zu gestalten, wurde ein Standarddialogobjekt (*CommonDialog*) in das Formular aufgenommen, so dass jede .avi-Datei auf Ihrem System abgespielt werden kann. Falls Sie keine .avi-Datei zur Hand haben, können Sie die Datei *Michael.avi* aus dem Ordner `\Vb6SfS\Lekt17` verwenden.

Das Programm RunVideo ausführen

- 1 Öffnen Sie das Projekt *RunVideo.vbp* aus dem Ordner `\Vb6SfS\Lekt17`.

- 2 Klicken Sie in der Symbolleiste auf die Schaltfläche *Starten*, um das Programm auszuführen.

Visual Basic zeigt das Formular der Anwendung an (siehe Abbildung 17.3).

- 3 Klicken Sie im Formular auf die Schaltfläche *.avi-Datei öffnen*, und verwenden Sie das Dialogfeld *Öffnen*, um eine geeignete .avi-Datei auf Ihrem System zu suchen und zu öffnen.

Wenn Sie möchten, können Sie die Videodatei *Michael.avi* aus dem Ordner `|Vb6SfS\Lekt17` laden.

- 4 Klicken Sie im Formular auf die Schaltfläche *.avi-Datei abspielen*, um das ausgewählte Video abzuspielen.

Wie in der folgenden Abbildung 17.4 gezeigt, wird ein zweites Fenster geöffnet, in dem das Video wiedergegeben wird. Wenn Sie sich mein Video (ein Ausschnitt aus meinem Programmierkurs *Learn Microsoft Visual Basic 6.0 Now*) ansehen, hören Sie mich auch sprechen. Probieren Sie nun einige Schaltflächen des Multimedia-MCI-Steuerelements aus, während das Video abgespielt wird.

- 5 Verschieben Sie zuerst das Fenster, in dem das Video abgespielt wird, falls es das Formular mit den Schaltflächen verdeckt.

- 6 Klicken Sie dann auf die Schaltfläche *Rücklauf* links unten im Multimedia-MCI-Steuerelement.

Das Video wird erneut gestartet. (Klicken Sie auf die Schaltfläche *Start*, falls dies nicht der Fall ist.)

RunVideo.

- 7 Klicken Sie auf die Schaltfläche *Pause*, um die Videowiedergabe zu unterbrechen.

- 8 Klicken Sie auf *Start*, um das Video wieder zu starten.

- 9 Klicken Sie auf die Schaltfläche *Stop*, um die Videowiedergabe zu beenden.

- a Klicken Sie auf die Schaltflächen *Zurück* und *Schritt*, um das Video um ein Bild vor bzw. zurück zu spulen.

- b Dann klicken Sie wieder auf die Schaltfläche *Start*, um das Video bis zum Schluß abzuspielen.

Wie Sie sehen, funktionieren die Schaltflächen des Multimedia-MCI-Steuerelements beim Abspielen von .avi-Dateien ähnlich wie die Tasten auf Ihrem Videorecorder.

- c Klicken Sie im Videofenster auf die Schaltfläche *Schließen*, um es zu schließen.

- d Wenn Sie möchten, können Sie mit Hilfe der Schaltflächen *.avi-Datei öffnen* und *.avi-Datei abspielen* weitere .avi-Dateien abspielen.

- e Schließlich klicken Sie im Formular auf die Schaltfläche *Beenden*, um das Programm zu beenden.

Der Programmcode zum Abspielen von .avi-Dateien

Eine der deutlichsten Unterschiede zwischen den beiden in dieser Lektion beschriebenen Programmen ist die Rolle, die das Multimedia-MCI-Steuererelement in der Benutzeroberfläche der Programme spielt. Im Programm *PlayTune* wurde das Steuererelement ausgeblendet, indem seine Eigenschaft *Visible* auf *False* eingestellt wurde. Im Programm *RunVideo* wird das Multimedia-MCI-Steuererelement angezeigt, da seine Eigenschaft *Visible* auf *True* (Voreinstellung) eingestellt wurde. Hier hat der Anwender Zugriff auf die Schaltflächen des Steuererelements.

Im folgenden Abschnitt wollen wir uns die Ereignisprozeduren im Programm *RunVideo* näher ansehen.

1 Öffnen Sie die Ereignisprozedur *Form_Load* im Codefenster.

Wie das Programm *PlayTune* beginnt das Programm *RunVideo* mit der Initialisierung des Multimedia-MCI-Steuererelements in der Ereignisprozedur *Form_Load*. In diesem Fall wird die Eigenschaft *DeviceType* aber auf *AVIVideo* eingestellt, um das Steuererelement für .avi-Dateien zu konfigurieren:

```
Private Sub Form_Load()  
MMControl1.Notify = False  
MMControl1.Wait = True  
MMControl1.Shareable = False  
MMControl1.DeviceType = "AVIVideo"  
End Sub
```

2 Öffnen Sie die Ereignisprozedur *cmdOpen_Click* im Codefenster.

Die Ereignisprozedur *cmdOpen_Click* wird jedes Mal ausgeführt, wenn der Anwender im Formular auf die Schaltfläche *.avi-Datei öffnen* klickt. Die Prozedur verwendet ein Standarddialogobjekt (*CommonDialog*), um das Dialogfeld *Öffnen* anzuzeigen, sowie die Eigenschaft *Filter*, damit lediglich .avi-Dateien im Dialogfeld *Öffnen* angezeigt werden. Wenn der Anwender auf die Schaltfläche *Abbrechen* klickt, beendet die Anweisung *Errhandler*: den Ladevorgang, indem die Anweisungen für das Multimedia MCI-Steuererelement übersprungen werden und die Anweisung *End* ausgeführt wird. Sobald der Anwender im Dialogfeld *Öffnen* aber eine gültige .avi-Datei auswählt, öffnet das Multimedia-MCI-Steuererelement diese Datei sofort, indem es der Eigenschaft *FileName* den Pfadnamen der Datei zuweist und mit der Eigenschaft *Command* den Befehl *Open* ausgibt:

```
Private Sub cmdOpen_Click()  
CommonDialog1.CancelError = True  
On Error GoTo Errhandler:  
CommonDialog1.Flags = cdIOFNFileMustExist  
CommonDialog1.Filter = "Video (*.AVI)|*.AVI"  
CommonDialog1.ShowOpen  
MMControl1.FileName = CommonDialog1.FileName  
MMControl1.Command = "Open"
```

Errhandler:

`Falls der Anwender auf Abbrechen klickt, Prozedur verlassen

End Sub

3 Öffnen Sie die Ereignisprozedur *cmdPlay_Click* im Codefenster.

Wenn der Anwender im Formular auf die Schaltfläche *.avi-Datei abspielen* klickt, gibt die Ereignisprozedur *cmdPlay_Click* den Befehl *Play* aus, woraufhin die .avi-Datei in einem eigenen Fenster wiedergegeben wird.

```
Private Sub cmdPlay_Click()  
MMControl1.Command = "Play"  
End Sub
```

4 Öffnen Sie die Ereignisprozedur *cmdQuit_Click* im Codefenster. Die Ereignisprozedur *cmdQuit_Click* beendet das Programm und entfernt das Formular aus dem Arbeitsspeicher. Diese Operation führt zur Ausführung der Ereignisprozedur *Form_Unload*, mit der das geöffnete Multimedia-Gerät geschlossen wird und die von diesem Gerät belegten Systemressourcen wieder freigegeben werden. Mit dem Befehl *Play* wird das Video gestartet.

```
Private Sub cmdQuit_Click()  
End  
End Sub
```

```
.  
. .
```

```
Private Sub Form_Unload(Cancel As Integer)  
MMControl1.Command = "Close"  
End Sub
```

Mit den oben beschriebenen Arbeitsschritten können Sie Videoclips, die im Dateiformat .avi gespeichert sind, in Ihre Anwendungen einbauen. Videoclips eignen sich zur Einbindung von Animationen, schrittweisen Anweisungen, Hilfeinformationen oder einfach dazu, einem Programm eine besondere Note zu verleihen.

Einen Schritt weiter: Musik von Audio-CDs wiedergeben

Im letzten Teil dieser Lektion werden Sie ein weiteres Leistungsmerkmal des Multimedia-MCI-Steuerelements anhand des Programms *PlayCD* kennen lernen. Dieses Programm spielt Musik ab, die auf Audio-CDs gespeichert ist. Programme mit einer musikalischen Untermalung zu versehen, hat offenkundige Vorteile. Sie müssen jedoch beachten, dass die hier gezeigte Vorgehensweise erfordert, dass sich eine Audio-CD im CDROM-Laufwerk Ihres Systems befindet. Für manche Anwender könnte sich diese Anforderung als nicht erfüllbar oder unpraktisch erweisen.

Das Programm *PlayCD* ausführen

- 1 Öffnen Sie das Projekt *PlayCD.vbp* aus dem Ordner `\Vb6SfS\Lekt17`.
- 2 Legen Sie eine Audio-CD in das (primäre) CD-ROM-Laufwerk Ihres Computers ein.
- 3 Warten Sie einen Augenblick, um zu sehen, ob die CD automatisch abgespielt wird.

Auf den meisten Systemen erkennt Windows automatisch, dass eine Audio-CD eingelegt wurde, und spielt sie mit dem mitgelieferten Programm *CD-Wiedergabe (cdplayer.exe)* ab. Wenn dies der Fall ist, schließen Sie das Programm *CD-Wiedergabe*, damit Sie das Programm *PlayCD* ausprobieren können.

4 Klicken Sie in der Symbolleiste auf die Schaltfläche *Starten*, um das Programm *PlayCD* in Visual Basic auszuführen.

Das Formular *PlayCD* mit dem Multimedia-MCI-Steuerelement wird auf dem Bildschirm angezeigt.

5 Klicken Sie im Formular auf die Schaltfläche *CD abspielen*, um das Audio-CD-Gerät (*CDAudio*) mit dem Multimedia-MCI-Steuerelement zu öffnen.

Im Formular erscheint ein roter Pfeil und die Anweisung *Hier klicken, um zu starten!*. Mit dieser Anweisung wird der Anwender darauf hingewiesen, dass die Wiedergabe durch Klicken auf die Schaltfläche *Start* gestartet wird. Der Pfeil und der Text wurden mit dem zum Lieferumfang von Windows gehörenden Programm *Paint* erstellt. Dann wurde die *Picture*-Eigenschaft des Ereignisses *cmdPlay_Click* verwendet, um die Bitmap-Grafik in ein Anzeigeobjekt (*Image*) des Formulars zu laden. Außerdem wird die *Click*-Ereignisprozedur der Schaltfläche *CD abspielen* verwendet, um den Inhalt des Anzeigefelds anzuzeigen.

6 Klicken Sie auf die Schaltfläche *Start* des Multimedia-MCI-Steuerelements. Die im CD-ROM-Laufwerk befindliche Audio-CD wird abgespielt.

7 Verwenden Sie die Schaltflächen des Multimedia-MCI-Steuerelements, um sich verschiedene Lieder der CD anzuhören.

8 Klicken Sie schließlich im Formular auf die Schaltfläche *Beenden*. Die Musik verstummt und das Programm wird geschlossen.

Der Programmcode des Programms *PlayCD*

1 Öffnen Sie die Ereignisprozedur *cmdPlay_Click* im Codefenster. Die Ereignisprozedur *cmdPlay_Click* wird ausgeführt, wenn der Anwender im Formular auf die Schaltfläche *CD abspielen* klickt. Anstatt das *CDAudio*-Gerät in der Ereignisprozedur *Form_Load* zu laden, wird dem Anwender die Gelegenheit gegeben, die Anweisungen auf dem Bildschirm durchzulesen und eine CD in das Laufwerk einzulegen, nachdem das Programm gestartet wurde. Sie sollten besonders zwei interessante Komponenten dieser Prozedur beachten. Erstens erfordert der hier angegebene Gerätetyp (*CDAudio*) keinen Dateinamen zur Angabe der abzuspielenden CD. Stattdessen wählt das *CDAudio*-Gerät einfach das primäre CD-ROM-Laufwerk im System aus und beginnt mit der Wiedergabe des ersten Liedes, sobald der Befehl *Play* gegeben wird. Obwohl einige neuere Audio-CDs Dateinameninformationen enthalten, sind diese für das aktuell geöffnete *CDAudio*-Gerät nicht notwendig. Zweitens verdient die letzte Zeile der Ereignisprozedur, in der die Eigenschaft *Visible* des Objekts *Image1* auf *True* gesetzt wird, besondere Beachtung. Mit dieser Programmanweisung wird der Pfeil und der Text *Hier klicken, um zu starten!* im Formular angezeigt.

```
Private Sub cmdPlay_Click()  
MMControl1.Notify = False  
MMControl1.Wait = True  
MMControl1.Shareable = False  
`CD-Audiotyp angeben (von CD-ROM-Laufwerk)  
MMControl1.DeviceType = "CDAudio"  
MMControl1.Command = "Open"  
Image1.Visible = True  
End Sub
```

2 Öffnen Sie die Ereignisprozedur *MMControl1_PlayClick* im Codefenster. Wie am Anfang dieser Lektion erwähnt, können Sie die Funktion der Schaltflächen des Multimedia-MCI-Steuerelements Ihren Wünschen entsprechend modifizieren, indem Sie Programmcode für die Schaltflächenereignisse dieses Steuerelements hinzufügen. In die folgende Ereignisprozedur wurde die Programmanweisung eingefügt, mit der der rote Pfeil und der Text (*Hier klicken, um zu starten!*) wieder ausgeblendet wird.


```
Private Sub MMControl1_PlayClick(Cancel As Integer)
```

```
'Vor der Wiedergabe Bitmap verbergen
```

```
Image1.Visible = False
```

```
End Sub
```

Beachten Sie, dass die Einstellung der Eigenschaft *Visible* von *Image1* den Befehl *Play* nicht ersetzt, der von der Schaltfläche *Start* gegeben wird. Sie wird einfach vor dem Befehl *Play* ausgeführt. Mit Ereignisprozeduren können Sie die Funktionen der Schaltflächen des Multimedia-MCI-Steuerelements modifizieren, ohne ihre Grundfunktionalität außer Kraft zu setzen. Wie Sie diese Möglichkeit nutzen, bleibt ganz Ihnen überlassen.

3 Öffnen Sie die Ereignisprozedur *cmdQuit_Click* im Codefenster. Mit der Befehlsschaltfläche *Beenden* wird die Wiedergabe der Audio-CD angehalten (falls sie gerade abgespielt wird) und das Programm mit der *End*-Anweisung beendet. Wie bereits oben beschrieben, bewirkt die *End*-Anweisung, dass die Ereignisprozedur *Form_Unload* ausgeführt und das geöffnete Multimedia-Gerät geschlossen wird.

```
Private Sub cmdQuit_Click()
```

```
'Wiedergabe der Audio-CD anhalten, wenn der Anwender auf Beenden klickt
```

```
MMControl1.Command = "Stop"
```

```
End
```

```
End Sub
```

```
Private Sub Form_Unload(Cancel As Integer)
```

```
'Gerät stets schließen, wenn die Wiedergabe beendet ist
```

```
MMControl1.Command = "Close"
```

```
End Sub
```

Sie haben nun gelernt, wie .wav-Dateien, .avi-Dateien und Audio-CDs mit den Multimedia-MCI-Steuerelement abgespielt werden. Diese drei Beispiele werden Ihnen helfen, entsprechende Multimedia-Funktionen in Ihre Programme zu integrieren.

Wenn Sie mit der nächsten Lektion fortfahren möchten

? Lassen Sie Visual Basic geladen, und schlagen Sie Lektion 18 auf.

Wenn Sie Visual Basic jetzt beenden möchten

? Klicken Sie im Menü *Datei* auf den Befehl *Beenden*.

Wenn daraufhin das Dialogfenster *Speichern* angezeigt wird, klicken Sie auf *Ja*.

Möchten Sie dann...

das Multimedia-MCI-ActiveXSteuerelement in die Visual Basic Werkzeugsammlung aufnehmen, klicken Sie im Menü *Projekt* auf den Befehl *Komponenten*. Klicken Sie auf das Register *Steuerelemente*, markieren Sie das Kontrollkästchen neben *Microsoft Multimedia Control 6.0*, und klicken Sie auf *OK*.

ein Multimedia-Gerät angeben, verwenden Sie die Eigenschaft *DeviceType* des Multimedia-MCI-Steuerelements.

Um beispielsweise das Wave-Audio-Gerät zum Abspielen von .wav-Dateien anzugeben, verwenden Sie folgende Anweisung:

```
MMControl1.DeviceType = "WaveAudio" P
```

einen MCI-Befehl an das angegebene Multimedia-Gerät schicken, verwenden Sie die Eigenschaft *Command*.

Um beispielsweise das durch die Eigenschaft *DeviceType* bezeichnete Multimedia-Gerät zu öffnen und abzuspielen, verwenden Sie folgende Anweisung:

```
MMControl1.Command = "Open"
```

```
MMControl1.Command = "Play"
```

das Multimedia-MCI-Steuerelement während der Programmausführung ausblenden, stellen Sie die Eigenschaft *Visible* des Multimedia-MCI-Steuerelements im Eigenschaftfenster auf *False*.

das Multimedia-MCI-Steuerelement während der Programmausführung anzeigen, stellen Sie die Eigenschaft *Visible* des Multimedia-MCI-Steuerelements im Eigenschaftfenster oder über Programmcode auf *True*.

das Multimedia-MCI-Steuerelement schließen und die belegten Systemressourcen freigeben, verwenden Sie den Befehl *Close* (*Schließen*) mit der Eigenschaft *Command* (befindet sich normalerweise in der Ereignisprozedur *Form_Unload*).

Zum Beispiel:

```
MMControl1.Command = "Close"
```